

35

EXHIBIT A

#24

8088 Assembler Language Programming: The IBM PC

By

David C. Willen

and

Jeffrey I. Krantz

Computer Applications Unlimited

Howard W. Sams & Co., Inc.
4300 WEST 82ND ST. INDIANAPOLIS, INDIANA 46268 USA

Copyright © 1983 by David C. Willen and
Jeffrey I. Krantz

FIRST EDITION
FIRST PRINTING—1983

All rights reserved. No part of this book shall be reproduced, stored in a retrieval system, or transmitted by any means, electronic, mechanical, photocopying, recording, or otherwise, without written permission from the publisher. No patent liability is assumed with respect to the use of the information contained herein. While every precaution has been taken in the preparation of this book, the publisher assumes no responsibility for errors or omissions. Neither is any liability assumed for damages resulting from the use of the information contained herein.

International Standard Book Number: 0-672-22024-5
Library of Congress Catalog Card Number: 83-060163

Illustrated by: T. R. Emrick

Printed in the United States of America.

```
MOV AL,OFDH  
OUT 21H,AL
```

Keep in mind that the state of the *interrupt flag* within the 8088 will ultimately determine whether or not any interrupt signal is received.

The second 8259 programming action that we must be concerned with is the signaling of the end of an interrupt service routine. This is accomplished by sending the "end of interrupt" (EOI) command, represented by 20H, to the interrupt command register within the 8259. Coincidentally, this one-byte register is accessed via i/o port 20H. That is all there is to controlling the interrupt mechanism! A complete example will appear later in this chapter.

THE 8255 PROGRAMMABLE PERIPHERAL INTERFACE

The 8255 is a general-purpose i/o interface chip that can be configured in many different ways. It is used on the system board to support a variety of devices and signals. These include the keyboard, speaker, configuration switches, and several other signals.

The chip contains three ports, called PA, PB, and PC. These are mapped to i/o addresses 60H, 61H, and 62H, respectively. In addition, there is a one-byte command register on the chip, accessed via i/o address 63H. On power-up, the BIOS initializes this chip by sending a value of 99H to the command register. This configures the 8255 so that PA and PC are considered input ports and PB is considered an output port. The meaning of each port is defined in Fig. 5-4. Note that additional logic on the system board allows us to select alternate inputs to ports PA and PC by setting certain bits in output port PB. In addition, we can read back the last value that was written to port PB by performing an input operation on port PB.

Fig. 5-5 gives an example of how we might make use of this hardware to read the settings of the configuration switches. There are two configuration switches on the system board; each can be set manually to represent any one-byte value. They are normally set up to indicate the various hardware options installed in the Personal Computer system. If, for example, our program needed to know how many disk drives were attached to the system, it could examine the two high-order bits of switch 1. This is accomplished by the program instructions of Fig. 5-5. Note that to enable the configuration-switch information onto port PA, we must first set bit 7 of port PB.

THE KEYBOARD

The system board provides an interface to the Personal Computer keyboard via the interrupt mechanism and ports PA and PB of the 8255 chip. This hardware is normally supported and controlled by programs running in the BIOS so that we do not have to be concerned with it. We simply




Fig. 5-4. 8255 port allocations.

access the keyboard via BIOS INT 16H, as shown in the last chapter. By understanding the hardware, however, we can write our own keyboard-support software, with certain interesting advantages.

```

IN      AL,61H      ;GET PRESENT VALUE OF PORT PB
OR      AL,80H      ;FORCE BIT 7 ON
OUT     61H,AL      ;SET PORT PB BIT 7 = 1 :
                      CONFIG SW1 NOW GATED TO PORT PA
IN      AL,60H      ;READ PORT PA = CONFIG SWITCH 1
NOT     AL          ;INVERT BITS
MOV     CL,6        ;SET UP SHIFT AMOUNT
SHR     AL,CL       ;ISOLATE BITS 7,6 OF AL
; NOW AL = NUMBER OF DISK DRIVES ATTACHED TO SYSTEM,
; AS OBTAINED FROM CONFIG SWITCH 1, POSITIONS 8,7.

```

Fig. 5-5. Reading the configuration switches.

Within the keyboard itself is a small microprocessor that scans for and detects any change in state of the keys. This processor receives its basic power and clock signals from the system board. We can disable the clock signal going to the keyboard by setting bit 6 of port PB to 0. This will prevent the keyboard from operating. In addition, we can send an *acknowledge signal* to the keyboard by setting bit 7 of port PB to 1. To ensure that the keyboard is properly enabled, we must set bits 7 and 6 of port PB to 0 and 1, respectively. In this state, the keyboard will generate an interrupt signal (IRQ1) whenever any key is depressed or released. It will then transmit a one-byte *scan code* to the system board and wait for the acknowledge signal to be returned. The scan code will be a number between 1 and 83 that uniquely identifies which key changed state (there are 83 keys on the keyboard). The high-order bit (bit 7) of the scan code indicates whether the key was depressed or released. It will be 0 if the key was depressed, and it will be 1 if the key was released. Fig. 5-6 identifies the scan code associated with each key on the keyboard.

It is the responsibility of the keyboard-support software to detect the keyboard interrupt and to respond to it as follows. First, the scan-code value transmitted to the system board must be obtained by reading from 8255 port PA. Then, the acknowledge signal must be sent back to the keyboard by momentarily setting bit 7 of port PB. The scan code itself may be interpreted in any manner desired. Thus, the meaning of each key can be defined, or even dynamically changed, by software. A more important consideration, however, stems from the fact that the keyboard interrupt occurs *asynchronously* with respect to the main program running in the computer. What this means is that the striking of a key (and its subsequent handling by the keyboard-support software) can occur at any time, and it is totally independent of when the main program may wish to receive keyboard input. Our keyboard support routine is therefore required to *buffer*, or save, any keyboard input that it receives. To accomplish this, we employ a "first-in, first-out" buffer, also referred to as a *circular queue*.

FUNCTION KEYS (LEFT SECTION)		
F1—59	*F5*—63	*F9*—67
F2—60	*F6*—64	*F10*—68
F3—61	*F7*—65	
F4—62	*F8*—66	
LETTER, NUMBER, AND PUNCTUATION KEYS (CENTER SECTION)		
1—2	*Y*—21	*~*—40
2—3	*U*—22	*~*—41
3—4	*I*—23	*^*—43
4—5	*O*—24	*Z*—44
5—6	*P*—25	*X*—45
6—7	*J*—26	*C*—46
7—8	*I*—27	*V*—47
8—9	*A*—30	*B*—48
9—10	*S*—31	*N*—49
0—11	*D*—32	*M*—50
~—12	*F*—33	*~*—51
=—13	*G*—34	*~*—52
Q—16	*H*—35	*~*—53
W—17	*J*—36	*~* (PriSc)—55
E—18	*K*—37	SPACE BAR —57
R—19	*L*—38	
T—20	*; *—39	
NUMERIC KEYPAD AREA (LEFT SECTION)		
7—71	*5*—76	*3*—81
8—72	*6*—77	*0*—82
9—73	*+ *—78	*~*—83
*- *—74	*I*—79	
4—75	*2*—80	
CONTROL KEYS (CENTER AND LEFT SECTIONS)		
Esc —1	Tab —15	Right SHIFT —54
Backspace —14	Enter —28	Alt —56
Num Lock —69	Ctrl —29	Caps Lock —58
Scroll Lock —70	Left SHIFT —42	

Fig. 5-6. Keyboard scan codes (listed in decimal).

Scan codes received from the keyboard are converted into the appropriate ASCII character codes and then placed onto this queue. When the main program wishes to obtain keyboard input, it calls an auxiliary routine within the keyboard-support software. This routine takes the characters off the queue, *in the order in which they were received*, and passes them to the main program. The size of the queue determines the maximum number of characters that can be buffered at any time. This represents the number of keystrokes that you can "type ahead" of the main program.

In Fig. 5-7, a complete program that sets up and utilizes its own keyboard-support software is presented. The program is kept relatively simple by omitting features normally handled by BIOS keyboard support, such as upper/lower-case alphabetic, "shift" and "shift-lock" keys, and special control-key combinations. The main program consists of two parts. Part one modifies the interrupt-service-routine address table to point to our own keyboard interrupt routine. It is also responsible for initializing the necessary hardware interfaces by sending commands to the 8259 and 8255 chips. Once this has been accomplished, we enter part two, a simple loop that reads keyboard input and displays it on the screen. The other

```

00030 ; EXAMPLE OF CUSTOM KEYBOARD SUPPORT SOFTWARE
00040 ;
00050 STACK SEGMENT PARA STACK 'STACK'
00060 DB 256 DUP (0) ;256 BYTES OF STACK SPACE
00070 STACK ENDS
00080 ;
00090 DATA SEGMENT PARA PUBLIC 'DATA'
00100 BUFFER DB 10 DUP (0) ;TEN BYTE KEYBOARD BUFFER
00110 BUFPTR1 DW 0 ;POINTS TO START OF BUFFER
00120 BUFPTR2 DW 0 ;POINTS TO END OF BUFFER
00130 ; NOTE: WHEN BUFPTR1 = BUFPTR2 , THEN THE BUFFER IS EMPTY.
00140 ; SCANTABLE CONVERTS SCAN CODES RECEIVED FROM THE KEYBOARD
00150 ; INTO THEIR CORRESPONDING ASCII CHARACTER CODES:
00160 SCANTABLE DB 0,0,'1234567890-=',8,0
00170 DB 'QWERTYUIOP[]',0DH,0
00180 DB 'ASDFGHJKL;',0,0,0,0,0
00190 DB 'ZXCVBNM,./',0,0,0,0
00200 DB ' ',0,0,0,0,0,0,0,0,0,0,0,0,0
00210 DB '789-456+1230.'
00220 DATA ENDS
00230 ;
00240 CODE SEGMENT PARA PUBLIC 'CODE'
00250 START PROC FAR
00260 ;
00270 ; STANDARD PROGRAM PROLOGUE
00280 ;
00290 ASSUME CS:CODE
00300 PUSH DS ;SAVE PSP SEG ADDR
00310 MOV AX,0
00320 PUSH AX ;SAVE RET ADDR OFFSET (PSP+0)
00330 MOV AX,DATA
00340 MOV DS,AX ;ESTABLISH DATA SEG ADDRESSABILITY
00350 ASSUME DS:DATA
00360 ;
00370 ; PART1: SETUP OUR OWN KEYBOARD INTERRUPT SERVICE ROUTINE
00380 ;
00390 CLI ;DISABLE ALL INTERRUPTS
00400 MOV AX,0
00410 MOV ES,AX ;POINT EXTRA SEGMENT AT THE...
00420 ; ...INTERRUPT SERVICE ROUTINE ADDRESS TABLE
00430 MOV DI,24H ;OFFSET OF ENTRY FOR TYPE CODE 09H
00440 MOV AX,OFFSET KBINT ;OFFSET OF OUR SERVICE ROUTINE
00445 CLD ;SET 'FORWARD' STRING OPERATIONS
00450 STOSW ;PLACE IT IN THE TABLE
00460 MOV AX,CS ;SEG OF OUR SERVICE ROUTINE
00470 STOSW ;PLACE IT IN THE TABLE
00480 MOV AL,0FCH ;ENABLE TIMER AND KYBD IRUPTS
00490 OUT 21H,AL ;WRITE INTERRUPT MASK REGISTER
00500 STI ;ENABLE INTERRUPTS TO THE 8088
00510 ;

```

Fig. 5-7. Custom keyboard-support program.

Continued on next page

```

00520 ; PART2: READ FROM KEYBOARD AND DISPLAY CHARS ON SCREEN
00530 ;
00540 FOREVER: CALL    KBGET    ;WAIT FOR A CHARACTER FROM THE KEYBOARD
00550             PUSH     AX      ;SAVE THE CHARACTER
00560             CALL     DISPCHAR ;DISPLAY THE CHARACTER RECEIVED
00570             POP      AX      ;RESTORE THE CHARACTER
00580             CMP      AL,0DH   ;WAS IT A CARRIAGE RETURN?
00590             JNZ      FOREVER ;BRANCH IF NOT
00600             MOV      AL,0AH   ;YES IT WAS, WE MUST ALSO DISPLAY...
00610             CALL     DISPCHAR ;...A LINE FEED!
00620             JMP      FOREVER ;STAY IN THIS LOOP FOREVER
00630 ;
00640 ; CALL KBGET TO WAIT FOR A CHARACTER TO BE RECEIVED FROM
00650 ; THE KEYBOARD. THE CHARACTER IS RETURNED IN REG AL.
00660 KBGET PROC NEAR
00670     PUSH     BX              ;SAVE REGISTER BX
00680     CLI                      ;DISABLE INTERRUPTS
00690     MOV      BX,BUFPTR1      ;START OF BUFFER
00700     CMP      BX,BUFPTR2      ;IS BUFFER EMPTY?
00710     JNZ      KBGET2         ;->NO
00720     STI                      ;RE-ENABLE INTERRUPTS
00730     POP      BX              ;RESTORE REGISTER BX
00740     JMP      KBGET          ;WAIT UNTIL SOMETHING IN BUFFER
00750 ; THERE IS SOMETHING IN THE BUFFER, GET IT :
00760 KBGET2: MOV      AL,[BUFFER+BX] ;GET CHAR AT BUFFER START
00770     INC      BX              ;INCREMENT BUFFER START
00780     CMP      BX,10          ;HAVE WE WRAPPED AROUND?
00790     JC       KBGET3         ;BRANCH IF NOT
00800     MOV      BX,0           ;YES, WRAP AROUND
00810 KBGET3: MOV      BUFPTR1,BX ;INDICATE NEW START OF BUFFER
00820     STI                      ;RE-ENABLE INTERRUPTS
00830     POP      BX              ;RESTORE REGISTER BX
00840     RET                      ;RETURN FROM KBGET
00850 KBGET ENDP
00860 ;
00870 ; KBINT IS OUR OWN KEYBOARD INTERRUPT SERVICE ROUTINE:
00880 ;
00890 KBINT PROC FAR
00900     PUSH     BX              ;SAVE ALL ALTERED REGISTERS!!
00910     PUSH     AX
00920 ;
00930 ; READ THE KEYBOARD DATA AND SEND THE ACKNOWLEDGE SIGNAL:
00940 ;
00950     IN       AL,60H          ;READ KEYBOARD INPUT
00960     PUSH     AX              ;SAVE KEYBOARD INPUT
00970     IN       AL,61H          ;READ 8255 PORT PB
00980     OR       AL,80H          ;SET KEYBOARD ACKNOWLEDGE SIGNAL
00990     OUT      61H,AL          ;SEND KEYBOARD ACKNOWLEDGE SIGNAL
01000     AND     AL,7FH          ;RESET KEYBOARD ACKNOWLEDGE SIGNAL
01010     OUT      61H,AL          ;RESTORE ORIGINAL 8255 PORT PB
01020 ;

```

Fig. 5-7 (cont). Custom keyboard-support program.

Continued on next page

Continued on next page.

```

01030 ; DECODE THE SCAN CODE RECEIVED:
01040 ;
01050     POP     AX      ;REGAIN THE KEYBOARD INPUT (AL)
01060     TEST    AL,80H  ;IS IT A KEY BEING RELEASED?
01070     JNZ     KBINT2  ;BRANCH IF YES, WE IGNORE THESE
01080     MOV     BX,OFFSET SCANTABLE ;SCAN CODE - ASCII TABLE
01090     XLATB          ;CONVERT THE SCAN CODE TO AN ASCII CHAR
01100     CMP     AL,0    ;IS IT A VALID ASCII KEY?
01110     JZ      KBINT2  ;BRANCH IF NOT
01120 ;
01130 ; PLACE THE ASCII CHARACTER INTO THE BUFFER:
01140 ;
01150     MOV     BX,BUFPTR2 ;GET POINTER TO END OF BUFFER
01160     MOV     [BUFFER+BX],AL ;PLACE CHAR IN BUFFER AT END
01170     INC     BX        ;INCREMENT BUFFER END
01180     CMP     BX,10     ;HAVE WE WRAPPED AROUND?
01190     JC      KBINT3    ;BRANCH IF NOT
01200     MOV     BX,0      ;YES, WRAP AROUND
01210 KBINT3: CMP     BX,BUFPTR1 ;IS BUFFER FULL?
01220     JZ      KBINT2    ;BRANCH IF YES, WE LOSE THIS CHAR
01230     MOV     BUFPTR2,BX ;INDICATE NEW END OF BUFFER
01240 ;
01250 ; NOW INDICATE "END OF INTERRUPT" TO THE INTERRUPT CONTROLLER:
01260 ;
01270 KBINT2: MOV     AL,20H ;SEND "EOI" COMMAND...
01280     OUT     20H,AL    ;...TO 8259 COMMAND REGISTER
01290     POP     AX      ;RESTORE ALL ALTERED REGISTERS!!
01300     POP     BX
01310     IRET          ;RETURN FROM INTERRUPT
01320 KBINT     ENDP
01330 ;
01340 ; SUBROUTINE TO DISPLAY A CHARACTER ON THE SCREEN.
01350 ; ENTER WITH AL = CHARACTER TO BE DISPLAYED.
01360 ; USES VIDEO INTERFACE IN BIOS.
01370 ;
01380 DISPCHAR PROC     NEAR
01390     PUSH     BX      ;SAVE BX REGISTER
01400     MOV     BX,0     ;SELECT DISPLAY PAGE 0
01410     MOV     AH,14    ;FUNCTION CODE FOR 'WRITE'
01420     INT     10H     ;CALL VIDEO DRIVER IN BIOS
01430     POP     BX      ;RESTORE BX REGISTER
01440     RET          ;RETURN TO CALLER OF 'DISPCHAR'
01450 DISPCHAR ENDP
01460 ;
01470 START     ENDP
01480 CODE     ENDS
01490     END      START

```

Fig. 5-7 (cont). Custom keyboard-support program.

major component of the program is our custom keyboard-support software. This also consists of two parts: they are KBINT, the keyboard interrupt-service routine, and KBGET, called from the main program to obtain keyboard input.

Let us look at the program in more detail. Statements 400 through 470 set the address of our own keyboard interrupt-service routine (KBINT) into the appropriate entry in the interrupt-service-routine address table. Recall that the keyboard interrupt signal is sent to the IRQ1 input of the 8259. The 8259 has been programmed to identify this interrupt source with a type code of 09H. The correct address-table entry therefore begins at physical address $09H \times 4$, or 00024H. Note that we disable interrupts (CLI) before altering the data in the address table. A catastrophic error could occur if an interrupt were to be received while the address table is being modified. Once the address table is modified, we program the interrupt-mask register of the 8259 to allow interrupts only from lines IRQ0 and IRQ1 (the timer and the keyboard, respectively). We then enable interrupts (STI) and enter the second part of the main program.

The second part (statements 540 through 620) is an infinite loop that calls routine KBGET to obtain characters input from the keyboard. Each character so received is echoed to the display screen by the DISPCHAR routine that we developed in the last chapter. Note the special code provided to detect the ENTER key (ASCII carriage return). This is necessary because a carriage return sent to an output device should always be followed by a line feed. If this is not done, we will find ourselves typing over the previous line of text.

If we strike a key while this loop is running, a type 09H interrupt will occur. This will cause our KBINT procedure to be activated. As you may recall, the 8088 interrupt response will also save the address of the instruction that was executing, save the flags, and disable further interrupts. The first responsibility of KBINT is to save any additional registers that it will use in servicing the interrupt (statements 900 and 910). It then reads in the scan code of the key that was depressed and sends back the acknowledge signal (statements 950 through 1010). If the scan code indicates that a key was being released (bit 7 = 1), then no further action is taken (statements 1060 and 1070). Otherwise, the XLATB instruction is used to convert the scan code into its corresponding ASCII character. The XLATB instruction requires that BX point to a translation table in the data segment. We therefore load BX with the offset address of SCANTABLE, which we have defined in our data segment. For each keyboard scan code that we wish to acknowledge, we have placed the appropriate ASCII code value into the corresponding position in SCANTABLE. Scan codes that we wish to ignore, such as those assigned to the function keys, F1–F10, are translated into a value of zero. After the translation, we test for a value of

zero. If we have such a value, then the key is ignored (statements 1100 and 1110).

Assuming a valid key has been struck, we now have its ASCII code in the AL register. We must place this byte onto the circular queue so that it is available to the main program. This is accomplished by statements 1150 through 1230. The queue itself is defined in the data segment, with the name BUFFER. It has the capacity to hold up to ten keystrokes. Two pointers, named BUFPTR1 and BUFPTR2, are used to keep track of the data in the queue. They point to the beginning and end of the valid data in the queue, respectively. Data is added onto the queue by placing it at the position pointed to by BUFPTR2, and then incrementing BUFPTR2. Data is taken off the queue by removing it from the position pointed to by BUFPTR1, and then incrementing BUFPTR1. When both pointers are equal, this indicates that there is no data in the queue. When incremented past the end of the queue, each pointer "wraps around" back to the beginning of the queue. This approach, illustrated in Fig. 5-8, ensures that we always retrieve data from the queue in the same order in which it was placed onto the queue. Notice that, in our implementation, we simply ignore (lose) a character if it is received when the queue is full.

Once the data has been placed onto the queue, we complete the interrupt response by sending the "end of interrupt" signal to the 8259 (statements 1270 and 1280). We then restore all saved registers and return to the main program, at its point of interruption, via an IRET instruction.

The main program relies on the KBGET routine (statements 640 through 850) to retrieve keyboard data from the circular queue. This routine waits until there is some data in the queue (as indicated by BUFPTR1 not equal to BUFPTR2). It then fetches that data, advances BUFPTR1, and returns the data value to its caller. Note that we must disable interrupts while the queue pointers are being manipulated. If this is not done, a keyboard inter-




Fig. 5-8. Circular-queue operations.

rupt may occur while we are trying to take data off the queue. We cannot allow data to be placed onto the queue at the same time that it is being taken off the queue, because this could cause us to overlook a queue-full condition.

If you type in this program, assemble it, and run it, you will be able to type most characters on the keyboard and have them echoed on the display screen. The only control keys that will function are "Backspace" and "Enter." Most other control keys will be ignored. Most important, however, is the fact that the control-key combinations CONTROL-BREAK and CONTROL-ALT-DEL are totally disabled. These functions are normally detected by the BIOS keyboard support. Since we have not provided such detection in our own program, we have effectively "locked up" the machine; the only way to exit from our program is to *turn the machine off*. This demonstrates the power and control that an assembler-language programmer can exert over his computer.

THE 8253 TIMER

The 8253 Timer chip can perform a number of different timing and/or counting functions. Within the chip are three independent counters, numbered 0, 1, and 2. Each of these three *timer channels* can be programmed to operate in one of six different modes, referred to as mode 0 through mode 5. Once they have been programmed, all of the channels can perform their designated counting or timing operations simultaneously. As you can imagine, some very sophisticated operations can be performed with this device.

A block diagram of the 8253 is presented in Fig. 5-9. Note that the hardware related to each timer channel is identical. Each channel contains a 16-bit *latch* register and a 16-bit *counter* register. Each channel also has two dedicated input signals, called *clock* and *gate*, as well as an output signal, *out*. In general, we program a count value into the latch register. From there, it is transferred into the counter register. Each time a pulse appears on the clock input, the value in the counter register is decremented by one. When the counter register reaches zero, a signal is generated on the out line. The mode to which we program the timer channel will determine exactly how each of these operations takes place.

The 8253 is programmed by writing commands into its one-byte-wide *command register*. In addition, each channel has a dedicated, one-byte-wide i/o port that is used to read and write the contents of its corresponding latch register. The i/o-port addresses used by the 8253 on the system board are listed in Table 5-2 and are shown in parentheses on Fig. 5-9.

The programming of a timer channel is always initiated by writing a command to the command-register port, 43H. The format of this command

8088 Assembler Language Programming: The IBM PC

By

David C. Willen

and

Jeffrey I. Krantz

Computer Applications Unlimited

Howard W. Sams & Co., Inc.
4300 WEST 62ND ST. INDIANAPOLIS, INDIANA 46268 USA

Copyright © 1983 by David C. Willen and
Jeffrey I. Krantz

FIRST EDITION
FIRST PRINTING—1983

All rights reserved. No part of this book shall be reproduced, stored in a retrieval system, or transmitted by any means, electronic, mechanical, photocopying, recording, or otherwise, without written permission from the publisher. No patent liability is assumed with respect to the use of the information contained herein. While every precaution has been taken in the preparation of this book, the publisher assumes no responsibility for errors or omissions. Neither is any liability assumed for damages resulting from the use of the information contained herein.

International Standard Book Number: 0-672-22024-5
Library of Congress Catalog Card Number: 83-060163

Illustrated by: T. R. Emrick

Printed in the United States of America.

```
MOV AL,0FDH
OUT 21H,AL
```

Keep in mind that the state of the *interrupt flag* within the 8088 will ultimately determine whether or not any interrupt signal is received.

The second 8259 programming action that we must be concerned with is the signaling of the end of an interrupt service routine. This is accomplished by sending the "end of interrupt" (EOI) command, represented by 20H, to the interrupt command register within the 8259. Coincidentally, this one-byte register is accessed via i/o port 20H. That is all there is to controlling the interrupt mechanism! A complete example will appear later in this chapter.

THE 8255 PROGRAMMABLE PERIPHERAL INTERFACE

The 8255 is a general-purpose i/o interface chip that can be configured in many different ways. It is used on the system board to support a variety of devices and signals. These include the keyboard, speaker, configuration switches, and several other signals.

The chip contains three ports, called PA, PB, and PC. These are mapped to i/o addresses 60H, 61H, and 62H, respectively. In addition, there is a one-byte command register on the chip, accessed via i/o address 63H. On power-up, the BIOS initializes this chip by sending a value of 99H to the command register. This configures the 8255 so that PA and PC are considered input ports and PB is considered an output port. The meaning of each port is defined in Fig. 5-4. Note that additional logic on the system board allows us to select alternate inputs to ports PA and PC by setting certain bits in output port PB. In addition, we can read back the last value that was written to port PB by performing an input operation on port PB.

Fig. 5-5 gives an example of how we might make use of this hardware to read the settings of the configuration switches. There are two configuration switches on the system board; each can be set manually to represent any one-byte value. They are normally set up to indicate the various hardware options installed in the Personal Computer system. If, for example, our program needed to know how many disk drives were attached to the system, it could examine the two high-order bits of switch 1. This is accomplished by the program instructions of Fig. 5-5. Note that to enable the configuration-switch information onto port PA, we must first set bit 7 of port PB.

THE KEYBOARD

The system board provides an interface to the Personal Computer keyboard via the interrupt mechanism and ports PA and PB of the 8255 chip. This hardware is normally supported and controlled by programs running in the BIOS so that we do not have to be concerned with it. We simply

```

IN      AL,61H      ;GET PRESENT VALUE OF PORT PB
OR      AL,80H      ;FORCE BIT 7 ON
OUT     61H,AL      ;SET PORT PB BIT 7 = 1 :
                     CONFIG SW1 NOW GATED TO PORT PA
IN      AL,60H      ;READ PORT PA = CONFIG SWITCH 1
NOT     AL          ;INVERT BITS
MOV     CL,6        ;SET UP SHIFT AMOUNT
SHR     AL,CL       ;ISOLATE BITS 7,6 OF AL
; NOW AL = NUMBER OF DISK DRIVES ATTACHED TO SYSTEM,
; AS OBTAINED FROM CONFIG SWITCH 1, POSITIONS 8,7.

```

Fig. 5-5. Reading the configuration switches.

Within the keyboard itself is a small microprocessor that scans for and detects any change in state of the keys. This processor receives its basic power and clock signals from the system board. We can disable the clock signal going to the keyboard by setting bit 6 of port PB to 0. This will prevent the keyboard from operating. In addition, we can send an *acknowledge signal* to the keyboard by setting bit 7 of port PB to 1. To ensure that the keyboard is properly enabled, we must set bits 7 and 6 of port PB to 0 and 1, respectively. In this state, the keyboard will generate an interrupt signal (IRQ1) whenever any key is depressed or released. It will then transmit a one-byte *scan code* to the system board and wait for the acknowledge signal to be returned. The scan code will be a number between 1 and 83 that uniquely identifies which key changed state (there are 83 keys on the keyboard). The high-order bit (bit 7) of the scan code indicates whether the key was depressed or released. It will be 0 if the key was depressed, and it will be 1 if the key was released. Fig. 5-6 identifies the scan code associated with each key on the keyboard.

It is the responsibility of the keyboard-support software to detect the keyboard interrupt and to respond to it as follows. First, the scan-code value transmitted to the system board must be obtained by reading from 8255 port PA. Then, the acknowledge signal must be sent back to the keyboard by momentarily setting bit 7 of port PB. The scan code itself may be interpreted in any manner desired. Thus, the meaning of each key can be defined, or even dynamically changed, by software. A more important consideration, however, stems from the fact that the keyboard interrupt occurs *asynchronously* with respect to the main program running in the computer. What this means is that the striking of a key (and its subsequent handling by the keyboard-support software) can occur at any time, and it is totally independent of when the main program may wish to receive keyboard input. Our keyboard support routine is therefore required to *buffer*, or *save*, any keyboard input that it receives. To accomplish this, we employ a "first-in, first-out" buffer, also referred to as a *circular queue*.

```

00030 ; EXAMPLE OF CUSTOM KEYBOARD SUPPORT SOFTWARE
00040 ;
00050 STACK SEGMENT PARA STACK 'STACK'
00060 DB 256 DUP (0) ;256 BYTES OF STACK SPACE
00070 STACK ENDS
00080 ;
00090 DATA SEGMENT PARA PUBLIC 'DATA'
00100 BUFFER DB 10 DUP (0) ;TEN BYTE KEYBOARD BUFFER
00110 BUFPTR1 DW 0 ;POINTS TO START OF BUFFER
00120 BUFPTR2 DW 0 ;POINTS TO END OF BUFFER
00130 ; NOTE: WHEN BUFPTR1 = BUFPTR2 , THEN THE BUFFER IS EMPTY.
00140 ; SCANTABLE CONVERTS SCAN CODES RECEIVED FROM THE KEYBOARD
00150 ; INTO THEIR CORRESPONDING ASCII CHARACTER CODES:
00160 SCANTABLE DB 0,0,'1234567890-='',8,0
00170 DB 'QWERTYUIOP[]',0DH,0
00180 DB 'ASDFGHJKL;',0,0,0,0,0
00190 DB 'ZXCVBNM,./',0,0,0,0
00200 DB ' ',0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
00210 DB '789-456+1230.'
00220 DATA ENDS
00230 ;
00240 CODE SEGMENT PARA PUBLIC 'CODE'
00250 START PROC FAR
00260 ;
00270 ; STANDARD PROGRAM PROLOGUE
00280 ;
00290 ASSUME CS:CODE
00300 PUSH DS ;SAVE PSP SEG ADDR
00310 MOV AX,0
00320 PUSH AX ;SAVE RET ADDR OFFSET (PSP+0)
00330 MOV AX,DATA
00340 MOV DS,AX ;ESTABLISH DATA SEG ADDRESSABILITY
00350 ASSUME DS:DATA
00360 ;
00370 ; PART1: SETUP OUR OWN KEYBOARD INTERRUPT SERVICE ROUTINE
00380 ;
00390 CLI ;DISABLE ALL INTERRUPTS
00400 MOV AX,0
00410 MOV ES,AX ;POINT EXTRA SEGMENT AT THE...
00420 ; ...INTERRUPT SERVICE ROUTINE ADDRESS TABLE
00430 MOV DI,24H ;OFFSET OF ENTRY FOR TYPE CODE 09H
00440 MOV AX,OFFSET KBINT ;OFFSET OF OUR SERVICE ROUTINE
00445 CLD ;SET 'FORWARD' STRING OPERATIONS
00450 STOSW ;PLACE IT IN THE TABLE
00460 MOV AX,CS ;SEG OF OUR SERVICE ROUTINE
00470 STOSW ;PLACE IT IN THE TABLE
00480 MOV AL,0FCH ;ENABLE TIMER AND KYBD INTERRUPTS
00490 OUT 21H,AL ;WRITE INTERRUPT MASK REGISTER
00500 STI ;ENABLE INTERRUPTS TO THE 8088
00510 ;

```

Fig. 5-7. Custom keyboard-support program.

Continued on next page.




Fig. 5-4. 8255 port allocations.

access the keyboard via BIOS INT 16H, as shown in the last chapter. By understanding the hardware, however, we can write our own keyboard-support software, with certain interesting advantages.

FUNCTION KEYS (LEFT SECTION)		
F1 —59	*F5* —63	*F9* —67
F2 —60	*F6* —64	*F10* —68
F3 —61	*F7* —65	
F4 —62	*F8* —66	
LETTER, NUMBER, AND PUNCTUATION KEYS (CENTER SECTION)		
1 —2	*Y* —21	*~* —40
2 —3	*U* —22	*" —41
3 —4	*I* —23	*\ —43
4 —5	*O* —24	*Z* —44
5 —6	*P* —25	*X* —45
6 —7	*J* —26	*C* —46
7 —8	*I* —27	*V* —47
8 —9	*A* —30	*B* —48
9 —10	*S* —31	*N* —49
0 —11	*D* —32	*M* —50
~ —12	*F* —33	*' —51
*_ —13	*G* —34	*" —52
Q —16	*H* —35	*/* —53
W —17	*J* —36	*** (PrtSc) —55
E —18	*K* —37	SPACE BAR —57
R —19	*L* —38	
T —20	*; —39	
NUMERIC KEYPAD AREA (LEFT SECTION)		
7 —71	*5* —76	*3* —81
8 —72	*6* —77	*0* —82
9 —73	*+ —78	*. —83
*_ —74	*1* —79	
4 —75	*2* —80	
CONTROL KEYS (CENTER AND LEFT SECTIONS)		
Esc —1	Tab —15	Right SHIFT —54
Backspace —14	Enter —28	Alt —56
Num Lock —69	Ctrl —29	Caps Lock —58
Scroll Lock —70	Left SHIFT —42	

Fig. 5-6. Keyboard scan codes (listed in decimal).

Scan codes received from the keyboard are converted into the appropriate ASCII character codes and then placed onto this queue. When the main program wishes to obtain keyboard input, it calls an auxiliary routine within the keyboard-support software. This routine takes the characters off the queue, *in the order in which they were received*, and passes them to the main program. The size of the queue determines the maximum number of characters that can be buffered at any time. This represents the number of keystrokes that you can "type ahead" of the main program.

In Fig. 5-7, a complete program that sets up and utilizes its own keyboard-support software is presented. The program is kept relatively simple by omitting features normally handled by BIOS keyboard support, such as upper/lower-case alphabets, "shift" and "shift-lock" keys, and special control-key combinations. The main program consists of two parts. Part one modifies the interrupt-service-routine address table to point to our own keyboard interrupt routine. It is also responsible for initializing the necessary hardware interfaces by sending commands to the 8259 and 8255 chips. Once this has been accomplished, we enter part two, a simple loop that reads keyboard input and displays it on the screen. The other

```

00520 ; PART2: READ FROM KEYBOARD AND DISPLAY CHARS ON SCREEN
00530 ;
00540 FOREVER: CALL KBGET ;WAIT FOR A CHARACTER FROM THE KEYBOARD
00550          PUSH AX ;SAVE THE CHARACTER
00560          CALL DISPCHAR ;DISPLAY THE CHARACTER RECEIVED
00570          POP AX ;RESTORE THE CHARACTER
00580          CMP AL,0DH ;WAS IT A CARRIAGE RETURN?
00590          JNZ FOREVER ;BRANCH IF NOT
00600          MOV AL,0AH ;YES IT WAS, WE MUST ALSO DISPLAY...
00610          CALL DISPCHAR ;...A LINE FEED!
00620          JMP FOREVER ;STAY IN THIS LOOP FOREVER
00630 ;
00640 ; CALL KBGET TO WAIT FOR A CHARACTER TO BE RECEIVED FROM
00650 ; THE KEYBOARD. THE CHARACTER IS RETURNED IN REG AL.
00660 KBGET PROC NEAR
00670          PUSH BX ;SAVE REGISTER BX
00680          CLI ;DISABLE INTERRUPTS
00690          MOV BX,BUFPTR1 ;START OF BUFFER
00700          CMP BX,BUFPTR2 ;IS BUFFER EMPTY?
00710          JNZ KBGET2 ;->NO
00720          STI ;RE-ENABLE INTERRUPTS
00730          POP BX ;RESTORE REGISTER BX
00740          JMP KBGET ;WAIT UNTIL SOMETHING IN BUFFER
00750 ; THERE IS SOMETHING IN THE BUFFER, GET IT :
00760 KBGET2: MOV AL,[BUFFER+BX] ;GET CHAR AT BUFFER START
00770          INC BX ;INCREMENT BUFFER START
00780          CMP BX,10 ;HAVE WE WRAPPED AROUND?
00790          JC KBGET3 ;BRANCH IF NOT
00800          MOV BX,0 ;YES, WRAP AROUND
00810 KBGET3: MOV BUFPTR1,BX ;INDICATE NEW START OF BUFFER
00820          STI ;RE-ENABLE INTERRUPTS
00830          POP BX ;RESTORE REGISTER BX
00840          RET ;RETURN FROM KBGET
00850 KBGET ENDP
00860 ;
00870 ; KBINT IS OUR OWN KEYBOARD INTERRUPT SERVICE ROUTINE:
00880 ;
00890 KBINT PROC FAR
00900          PUSH BX ;SAVE ALL ALTERED REGISTERS!!
00910          PUSH AX
00920 ;
00930 ; READ THE KEYBOARD DATA AND SEND THE ACKNOWLEDGE SIGNAL:
00940 ;
00950          IN AL,60H ;READ KEYBOARD INPUT
00960          PUSH AX ;SAVE KEYBOARD INPUT
00970          IN AL,61H ;READ 8255 PORT PB
00980          OR AL,80H ;SET KEYBOARD ACKNOWLEDGE SIGNAL
00990          OUT 61H,AL ;SEND KEYBOARD ACKNOWLEDGE SIGNAL
01000          AND AL,7FH ;RESET KEYBOARD ACKNOWLEDGE SIGNAL
01010          OUT 61H,AL ;RESTORE ORIGINAL 8255 PORT PB
01020 ;

```

Fig. 5-7 (cont). Custom keyboard-support program.

Continued on next page.

```

01030 ; DECODE THE SCAN CODE RECEIVED:
01040 ;
01050     POP     AX      ;REGAIN THE KEYBOARD INPUT (AL)
01060     TEST    AL,80H  ;IS IT A KEY BEING RELEASED?
01070     JNZ     KBINT2  ;BRANCH IF YES, WE IGNORE THESE
01080     MOV     BX,OFFSET SCANTABLE ;SCAN CODE - ASCII TABLE
01090     XLATB      ;CONVERT THE SCAN CODE TO AN ASCII CHAR
01100     CMP     AL,0     ;IS IT A VALID ASCII KEY?
01110     JZ      KBINT2  ;BRANCH IF NOT
01120 ;
01130 ; PLACE THE ASCII CHARACTER INTO THE BUFFER:
01140 ;
01150     MOV     BX,BUFPTR2 ;GET POINTER TO END OF BUFFER
01160     MOV     [BUFFER+BX],AL ;PLACE CHAR IN BUFFER AT END
01170     INC     BX      ;INCREMENT BUFFER END
01180     CMP     BX,10    ;HAVE WE WRAPPED AROUND?
01190     JC      KBINT3   ;BRANCH IF NOT
01200     MOV     BX,0     ;YES, WRAP AROUND
01210 KBINT3: CMP     BX,BUFPTR1 ;IS BUFFER FULL?
01220     JZ      KBINT2  ;BRANCH IF YES, WE LOSE THIS CHAR
01230     MOV     BUFPTR2,BX ;INDICATE NEW END OF BUFFER
01240 ;
01250 ; NOW INDICATE "END OF INTERRUPT" TO THE INTERRUPT CONTROLLER:
01260 ;
01270 KBINT2: MOV     AL,20H ;SEND "EOI" COMMAND...
01280     OUT     20H,AL    ;...TO 8259 COMMAND REGISTER
01290     POP     AX      ;RESTORE ALL ALTERED REGISTERS!!
01300     POP     BX
01310     IRET      ;RETURN FROM INTERRUPT
01320 KBINT     ENDP
01330 ;
01340 ; SUBROUTINE TO DISPLAY A CHARACTER ON THE SCREEN.
01350 ; ENTER WITH AL = CHARACTER TO BE DISPLAYED.
01360 ; USES VIDEO INTERFACE IN BIOS.
01370 ;
01380 DISPCHAR PROC     NEAR
01390     PUSH     BX      ;SAVE BX REGISTER
01400     MOV     BX,0     ;SELECT DISPLAY PAGE 0
01410     MOV     AH,14    ;FUNCTION CODE FOR 'WRITE'
01420     INT     10H     ;CALL VIDEO DRIVER IN BIOS
01430     POP     BX      ;RESTORE BX REGISTER
01440     RET      ;RETURN TO CALLER OF 'DISPCHAR'
01450 DISPCHAR ENDP
01460 ;
01470 START     ENDP
01480 CODE     ENDS
01490     END      START

```

Fig. 5-7 (cont). Custom keyboard-support program.

major component of the program is our custom keyboard-support software. This also consists of two parts; they are KBINT, the keyboard interrupt-service routine, and KBGET, called from the main program to obtain keyboard input.

Let us look at the program in more detail. Statements 400 through 470 set the address of our own keyboard interrupt-service routine (KBINT) into the appropriate entry in the interrupt-service-routine address table. Recall that the keyboard interrupt signal is sent to the IRQ1 input of the 8259. The 8259 has been programmed to identify this interrupt source with a type code of 09H. The correct address-table entry therefore begins at physical address $09H \times 4$, or 00024H. Note that we disable interrupts (CLI) before altering the data in the address table. A catastrophic error could occur if an interrupt were to be received while the address table is being modified. Once the address table is modified, we program the interrupt-mask register of the 8259 to allow interrupts only from lines IRQ0 and IRQ1 (the timer and the keyboard, respectively). We then enable interrupts (STI) and enter the second part of the main program.

The second part (statements 540 through 620) is an infinite loop that calls routine KBGET to obtain characters input from the keyboard. Each character so received is echoed to the display screen by the DISPCHAR routine that we developed in the last chapter. Note the special code provided to detect the ENTER key (ASCII carriage return). This is necessary because a carriage return sent to an output device should always be followed by a line feed. If this is not done, we will find ourselves typing over the previous line of text.

If we strike a key while this loop is running, a type 09H interrupt will occur. This will cause our KBINT procedure to be activated. As you may recall, the 8088 interrupt response will also save the address of the instruction that was executing, save the flags, and disable further interrupts. The first responsibility of KBINT is to save any additional registers that it will use in servicing the interrupt (statements 900 and 910). It then reads in the scan code of the key that was depressed and sends back the acknowledge signal (statements 950 through 1010). If the scan code indicates that a key was being released (bit 7 = 1), then no further action is taken (statements 1060 and 1070). Otherwise, the XLATB instruction is used to convert the scan code into its corresponding ASCII character. The XLATB instruction requires that BX point to a translation table in the data segment. We therefore load BX with the offset address of SCANTABLE, which we have defined in our data segment. For each keyboard scan code that we wish to acknowledge, we have placed the appropriate ASCII code value into the corresponding position in SCANTABLE. Scan codes that we wish to ignore, such as those assigned to the function keys, F1–F10, are translated into a value of zero. After the translation, we test for a value of

zero. If we have such a value, then the key is ignored (statements 1100 and 1110).

Assuming a valid key has been struck, we now have its ASCII code in the AL register. We must place this byte onto the circular queue so that it is available to the main program. This is accomplished by statements 1150 through 1230. The queue itself is defined in the data segment, with the name BUFFER. It has the capacity to hold up to ten keystrokes. Two pointers, named BUFPTR1 and BUFPTR2, are used to keep track of the data in the queue. They point to the beginning and end of the valid data in the queue, respectively. Data is added onto the queue by placing it at the position pointed to by BUFPTR2, and then incrementing BUFPTR2. Data is taken off the queue by removing it from the position pointed to by BUFPTR1, and then incrementing BUFPTR1. When both pointers are equal, this indicates that there is no data in the queue. When incremented past the end of the queue, each pointer "wraps around" back to the beginning of the queue. This approach, illustrated in Fig. 5-8, ensures that we always retrieve data from the queue in the same order in which it was placed onto the queue. Notice that, in our implementation, we simply ignore (lose) a character if it is received when the queue is full.

Once the data has been placed onto the queue, we complete the interrupt response by sending the "end of interrupt" signal to the 8259 (statements 1270 and 1280). We then restore all saved registers and return to the main program, at its point of interruption, via an IRET instruction.

The main program relies on the KBGET routine (statements 640 through 850) to retrieve keyboard data from the circular queue. This routine waits until there is some data in the queue (as indicated by BUFPTR1 not equal to BUFPTR2). It then fetches that data, advances BUFPTR1, and returns the data value to its caller. Note that we must disable interrupts while the queue pointers are being manipulated. If this is not done, a keyboard inter-




Fig. 5-8. Circular-queue operations.

rupt may occur while we are trying to take data off the queue. We cannot allow data to be placed onto the queue at the same time that it is being taken off the queue, because this could cause us to overlook a queue-full condition.

If you type in this program, assemble it, and run it, you will be able to type most characters on the keyboard and have them echoed on the display screen. The only control keys that will function are "Backspace" and "Enter." Most other control keys will be ignored. Most important, however, is the fact that the control-key combinations CONTROL-BREAK and CONTROL-ALT-DEL are totally disabled. These functions are normally detected by the BIOS keyboard support. Since we have not provided such detection in our own program, we have effectively "locked up" the machine; the only way to exit from our program is to *turn the machine off*. This demonstrates the power and control that an assembler-language programmer can exert over his computer.

THE 8253 TIMER

The 8253 Timer chip can perform a number of different timing and/or counting functions. Within the chip are three independent counters, numbered 0, 1, and 2. Each of these three *timer channels* can be programmed to operate in one of six different modes, referred to as mode 0 through mode 5. Once they have been programmed, all of the channels can perform their designated counting or timing operations simultaneously. As you can imagine, some very sophisticated operations can be performed with this device.

A block diagram of the 8253 is presented in Fig. 5-9. Note that the hardware related to each timer channel is identical. Each channel contains a 16-bit *latch* register and a 16-bit *counter* register. Each channel also has two dedicated input signals, called *clock* and *gate*, as well as an output signal, *out*. In general, we program a count value into the latch register. From there, it is transferred into the counter register. Each time a pulse appears on the clock input, the value in the counter register is decremented by one. When the counter register reaches zero, a signal is generated on the out line. The mode to which we program the timer channel will determine exactly how each of these operations takes place.

The 8253 is programmed by writing commands into its one-byte-wide *command register*. In addition, each channel has a dedicated, one-byte-wide i/o port that is used to read and write the contents of its corresponding latch register. The i/o-port addresses used by the 8253 on the system board are listed in Table 5-2 and are shown in parentheses on Fig. 5-9.

The programming of a timer channel is always initiated by writing a command to the command-register port, 43H. The format of this command